Worksheet: **`Lecture2.mws`**

Lecture 2:

Cooling of a Hot Bar: The Diffusion Equation

Andrew J. Bernoff

PCMI, Summer 2003

This worksheet contains the examples from the second lecture.

The Diffusion Equation: Solutions to the Dirchlect Problem

```
> restart:
> with(plots):
Warning, the name changecoords has been redefined
```

Some solutions to the diffusion equation for a metal bar of length π with ends held at zero temperature:

$U_t = \mathrm{D}\,U_{xx}$ 0<x<π , 0<t **DE**

$\mathrm{U}(0, t) = 0$ $\mathrm{U}(\pi, t) = 0$ 0<t **BC**

$\mathrm{U}(x, 0) = \mathrm{f}(x)$ 0<x<π **IC**

```
> K:=1;
```
Note "K" is the diffusion constant here !! MAPLE reserves "D" for other uses

$$K := 1$$

```
> Un:=(n,x,t)->sin(n*x)*exp(-(n^2)*K*t);
```

$$Un := (n, x, t) \rightarrow \sin(x\,n)\,\mathbf{e}^{(-n^2\,K\,t)}$$

```
> Ut:=diff(Un(n,x,t),t);Uxx:=diff(Un(n,x,t),x$2);Ut-Uxx;
```

$$Ut := -\sin(x\,n)\,n^2\,\mathbf{e}^{(-n^2\,t)}$$

$$Uxx := -\sin(x\,n)\,n^2\,\mathbf{e}^{(-n^2\,t)}$$

$$0$$

```
> animate(Un(1,x,t),x=0..Pi,t=0..3,frames=100,color=blue,thickness=2);
```




```
> animate(Un(2,x,t),x=0..Pi,t=0..3,frames=100,color=blue,thickness=2);
```




```
> U:=8/(Pi^2)*sum((-1)^k*Un((2*k+1),x,t)/(2*k+1)^2,k=0..20);
```

$$U := 8\left(\sin(x)\,\mathbf{e}^{(-t)} - \frac{1}{9}\sin(3\,x)\,\mathbf{e}^{(-9\,t)} + \frac{1}{25}\sin(5\,x)\,\mathbf{e}^{(-25\,t)} - \frac{1}{49}\sin(7\,x)\,\mathbf{e}^{(-49\,t)} + \frac{1}{81}\sin(9\,x)\,\mathbf{e}^{(-81\,t)} - \frac{1}{121}\sin(11\,x)\,\mathbf{e}^{(-121\,t)} + \frac{1}{169}\sin(13\,x)\,\mathbf{e}^{(-169\,t)} - \frac{1}{225}\sin(15\,x)\,\mathbf{e}^{(-225\,t)} + \frac{1}{289}\sin(17\,x)\,\mathbf{e}^{(-289\,t)} - \frac{1}{361}\sin(19\,x)\,\mathbf{e}^{(-361\,t)} + \frac{1}{441}\sin(21\,x)\,\mathbf{e}^{(-441\,t)} - \frac{1}{529}\sin(23\,x)\,\mathbf{e}^{(-529\,t)} + \frac{1}{625}\sin(25\,x)\,\mathbf{e}^{(-625\,t)} - \frac{1}{729}\sin(27\,x)\,\mathbf{e}^{(-729\,t)} + \frac{1}{841}\sin(29\,x)\,\mathbf{e}^{(-841\,t)} - \frac{1}{961}\sin(31\,x)\,\mathbf{e}^{(-961\,t)} + \frac{1}{1089}\sin(33\,x)\,\mathbf{e}^{(-1089\,t)} - \frac{1}{1225}\sin(35\,x)\,\mathbf{e}^{(-1225\,t)} + \frac{1}{1369}\sin(37\,x)\,\mathbf{e}^{(-1369\,t)} - \frac{1}{1521}\sin(39\,x)\,\mathbf{e}^{(-1521\,t)} + \frac{1}{1681}\sin(41\,x)\,\mathbf{e}^{(-1681\,t)}\right) \Big/ \pi^2$$

```
> animate(U,x=0..Pi,t=0..3,frames=100,color=blue,thickness=2);
```




```
> plot3d(U,x=0..Pi,t=0..3,style=patchnogrid,shading=ZHUE,axes=boxe
  d);
```



The Diffusion Equation: Solutions to the Neumann Problem

```
> restart:
> with(plots):
Warning, the name changecoords has been redefined
```

Now, let's insulate the ends of the bar for the same initial condition:

$U_t = D\,U_{xx}$ $\quad 0<x<\pi$, $\quad 0<t$ $\quad$ **DE**

$U_x(0,t)=0$ $\quad U_x(\pi,t)=0$ $\quad 0<t$ $\quad$ **BC**

$U(x,0)=f(x)$ $\quad 0<x<\pi$ $\quad$ **IC**

```
> K:=1;
```

$$K := 1$$

```
> Vn:=(n,x,t)->cos(n*x)*exp(-(n^2)*K*t);
```

$$Vn := (n, x, t) \to \cos(x\,n)\,\mathbf{e}^{(-n^2 K\,t)}$$

```
> Vt:=diff(Vn(n,x,t),t);Vxx:=diff(Vn(n,x,t),x$2);Vt-Vxx;
```

$$Vt := -\cos(x\,n)\,n^2\,\mathbf{e}^{(-n^2 t)}$$

$$Vxx := -\cos(x\,n)\,n^2\,\mathbf{e}^{(-n^2 t)}$$

$$0$$

```
> V:=1/2-16/(Pi^2)*sum(Vn((4*k+2),x,t)/(4*k+2)^2,k=0..20);
```

$$V := \frac{1}{2} - 16\left(\frac{1}{4}\cos(2\,x)\,\mathbf{e}^{(-4\,t)} + \frac{1}{36}\cos(6\,x)\,\mathbf{e}^{(-36\,t)} + \frac{1}{100}\cos(10\,x)\,\mathbf{e}^{(-100\,t)} + \frac{1}{196}\cos(14\,x)\,\mathbf{e}^{(-196\,t)} + \frac{1}{324}\cos(18\,x)\,\mathbf{e}^{(-324\,t)} + \frac{1}{484}\cos(22\,x)\,\mathbf{e}^{(-484\,t)} + \frac{1}{676}\cos(26\,x)\,\mathbf{e}^{(-676\,t)} + \frac{1}{900}\cos(30\,x)\,\mathbf{e}^{(-900\,t)} + \frac{1}{1156}\cos(34\,x)\,\mathbf{e}^{(-1156\,t)} + \frac{1}{1444}\cos(38\,x)\,\mathbf{e}^{(-1444\,t)} + \frac{1}{1764}\cos(42\,x)\,\mathbf{e}^{(-1764\,t)} + \frac{1}{2116}\cos(46\,x)\,\mathbf{e}^{(-2116\,t)} + \frac{1}{2500}\cos(50\,x)\,\mathbf{e}^{(-2500\,t)} + \frac{1}{2916}\cos(54\,x)\,\mathbf{e}^{(-2916\,t)} + \frac{1}{3364}\cos(58\,x)\,\mathbf{e}^{(-3364\,t)} + \frac{1}{3844}\cos(62\,x)\,\mathbf{e}^{(-3844\,t)} + \frac{1}{4356}\cos(66\,x)\,\mathbf{e}^{(-4356\,t)} + \frac{1}{4900}\cos(70\,x)\,\mathbf{e}^{(-4900\,t)} + \frac{1}{5476}\cos(74\,x)\,\mathbf{e}^{(-5476\,t)} + \frac{1}{6084}\cos(78\,x)\,\mathbf{e}^{(-6084\,t)} + \frac{1}{6724}\cos(82\,x)\,\mathbf{e}^{(-6724\,t)}\right)\Big/\pi^2$$

```
> animate(V,x=0..Pi,t=0..3,frames=100,color=blue,thickness=2);
```




```
> plot3d(V,x=0..Pi,t=0..1,style=patchnogrid,shading=ZHUE,axes=boxe
  d);
```



The Diffusion Equation: Solutions to the Cauchy Problem

```
> restart:
> with(plots):
Warning, the name changecoords has been redefined
```

Finally consider the problem on an infinite interval

$U_t = D\,U_{xx}$ $-\infty<x<\infty$, $0<t$ **DE**

$\lim_{x\to\infty} U(x,t) < c$ $\lim_{x\to(-\infty)} U(x,t) < c$ $0<t$ **BC**

$U(x,0) = f(x)$ $-\infty<x<\infty$ **IC**

```
> K:=1;tau:=1;
```

$$K := 1$$

$$\tau := 1$$

```
> G:=(x,t)-> 1/sqrt(4*Pi*K*(t+tau))*exp(-x^2/(4*K*(t+tau)));
```

$$G := (x,t) \to \frac{e^{\left(-1/4\,\frac{x^2}{K\,(t+\tau)}\right)}}{\sqrt{4\,\pi\,K\,(t+\tau)}}$$

```
> Gt:=diff(G(x,t),t);Gxx:=diff(G(x,t),x$2);simplify(Gt-K*Gxx);
```

$$Gt := -\frac{1}{4}\,\frac{e^{\left(-\frac{x^2}{4\,(t+1)}\right)}\,\pi}{(\pi\,(t+1))^{(3/2)}} + \frac{1}{8}\,\frac{x^2\,e^{\left(-\frac{x^2}{4\,(t+1)}\right)}}{\sqrt{\pi\,(t+1)}\,(t+1)^2}$$

$$Gxx := -\frac{1}{4}\,\frac{e^{\left(-\frac{x^2}{4\,(t+1)}\right)}}{\sqrt{\pi\,(t+1)}\,(t+1)} + \frac{1}{8}\,\frac{x^2\,e^{\left(-\frac{x^2}{4\,(t+1)}\right)}}{\sqrt{\pi\,(t+1)}\,(t+1)^2}$$

$$0$$

```
> animate(G(x,t),x=-10..10,t=0..10,numpoints=200,color=blue,thickn
  ess=2,title="Heat Kernel");
```




```
> plot3d(G(x,t),x=-10..10,t=0..10,style=patchnogrid,shading=ZHUE,a
  xes=boxed);
```